Don Taylor

Quaternionic Reflection Groups

Version: 5 October 2025
TEXed: 21 October 2025
File: QuatRefCode.tex

This file describes Magma code to construct irreducible finite quaternionic reflection groups. The code itself is in the file QuatRefCode.m.

1. Quaternionic cyclic groups

The cyclic group C_m generated by the m-th root of unity $\zeta_m = \cos 2\pi/m + i \sin 2\pi/m$.

```
qC := function(m)
  if m gt 2 then
    mm := IsEven(m) select m else 2*m;
    F<z> := CyclotomicField(2*mm);
    K<zz> := sub< F | z + z^-1 >;    // the real subfield
    _<t> := PolynomialRing(K);
    C<ii> := ext< K | t^2 + 1 >;
    ok, f := IsIsomorphic(F, C);
    assert ok;
    A<i,j,k> := QuaternionAlgebra< K | -1, -1 >;
    es := IsEven(m) select Eltseq(f(z)^2) else Eltseq(f(z)^4);
    ζ := es[1]+es[2]*i;
  else
    A<i,j,k> := QuaternionAlgebra< Rationals() | -1, -1 >;
    ζ := (m eq 1) select Id(A) else -Id(A);
  end if;
  return MatrixGroup< 1, A | [ζ] >;
end function;
```

A version of qC that also returns a square root of 2.

```
a2C := function(m)
  mm := case< m mod 4 | 0 : m, 2 : 2*m, default : 4*m>;
  F<z> := CyclotomicField(2*mm);
  q := mm div 4;
  a2 := z^-q + z^q;
  K<zz> := sub< F | z + z^-1 >;
  _<t> := PolynomialRing(K);
  C<ii> := ext< K | t^2 + 1 >;
  ok, f := IsIsomorphic(F, C);
  assert ok;
  A<i,j,k> := QuaternionAlgebra< K | -1, -1 >;
  es := case< m mod 4 | 0 : f(z)^2, 2 : f(z)^4, default : f(z)^8>;
  ζ := es[1]+es[2]*i;
```

```
    return MatrixGroup< 1, A | [ζ] >, a2;
  end function;
```

A version of qC that also returns a square root of 5.

```
  a5C := function(m)
    if m mod 5 ne 0 then
      mm := IsEven(m) select 5*m else 10*m;
    else
      mm := IsEven(m) select m else 2*m;
    end if;
    F<z> := CyclotomicField(2*mm);
    q := 2*mm div 5;
    ξ := z^q;
    a5 := -2*ξ^2 - 2*ξ^3 -1;
    K<zz> := sub< F | z + z^-1 >;
    _<t> := PolynomialRing(K);
    C<ii> := ext< K | t^2 + 1 >;
    ok, f := IsIsomorphic(F, C);
    assert ok;
    A<i,j,k> := QuaternionAlgebra< K | -1, -1 >;
    es := f(z)^(2*mm div m);
    ζ := es[1]+es[2]*i;
    return MatrixGroup< 1, A | [ζ] >, A ! a5;
  end function;
```

2. Binary dihedral groups

The binary dihedral group $\mathcal{D}_m$ of order $4m$ is generated by the $2m$-th root of unity ζ_{2m} and the quaternion j.

The binary dihedral group $\mathcal{D}_m$ of order $4m$.

```
  qD := function(m)
    if m gt 2 then
      mm := IsEven(m) select m else 2*m;
      F<z> := CyclotomicField(2*mm);
      K<zz> := sub< F | z + z^-1 >;
      _<t> := PolynomialRing(K);
      C<ii> := ext< K | t^2 + 1 >;
      ok, f := IsIsomorphic(F, C);
      assert ok;
      A<i,j,k> := QuaternionAlgebra< K | -1, -1 >;
      es := IsEven(m) select Eltseq(f(z)) else Eltseq(f(z)^2);
      ζ := es[1]+es[2]*i;
    else
```

```
    A<i,j,k> := QuaternionAlgebra< Rationals() | −1, −1 >;
    ζ := (m eq 1) select i^2 else i;
  end if;
  return MatrixGroup< 1, A | [ζ], [j] >;
end function;
```

A version of qD that also returns a square root of 2.

```
a2D := function(m)
  mm := case< m mod 4 | 0 : m, 6 : 2*m, default : 4*m>;
  F<z> := CyclotomicField(2*mm);
  q := mm div 4;
  a2 := z^-q + z^q;
  K<zz> := sub< F | z + z^-1 >;
  _<t> := PolynomialRing(K);
  C<ii> := ext< K | t^2 + 1 >;
  ok, f := IsIsomorphic(F, C);
  assert ok;
  A<i,j,k> := QuaternionAlgebra< K | −1, −1 >;
  es := case< m mod 4 | 0 : f(z), 6 : f(z)^2, default : f(z)^4>;
  ζ := es[1]+es[2]*i;
  return MatrixGroup< 1, A | [ζ], [j] >, a2;
end function;
```

3. The rank 2 quaternionic reflection groups $G(\mathcal{D}_m, H, \psi_r)$

Suppose that $2m = n\ell$ and let $H = C_\ell = \langle \zeta_{2m}^n \rangle$. Let ψ_r denote the automorphism of $\mathcal{D}_m/H$ such that $\psi_r(H\zeta_{2m}) = H\zeta_{2m}^r$ and $\psi_r(Hj) = Hj^{-1}$.

If $r = 1$ or $1 < r \leq n/2$ and $\gcd(r, n) = \gcd(\kappa, \nu) = 1$, where $\kappa = n/\gcd(n, r+1)$ and $\nu = n/\gcd(n, r-1)$, then $G(\mathcal{D}_m, C_\ell, \psi_r)$ is a quaternionic reflection with generators

$$\begin{pmatrix} 0 & 1 \\ 1 & 0 \end{pmatrix}, \quad \begin{pmatrix} \zeta_{2m}^n & 0 \\ 0 & 1 \end{pmatrix}, \quad \begin{pmatrix} \zeta_{2m} & 0 \\ 0 & \zeta_{2m}^r \end{pmatrix}, \quad \begin{pmatrix} j & 0 \\ 0 & j^{-1} \end{pmatrix}.$$

The list of valid parameters r for $G(\mathcal{D}_m, C_\ell, \psi_r)$. Note that 1 is always valid.

```
qValidParametersDC := function(m, ℓ)
  pp := [];
  if IsOdd(ℓ) then
    ok, m0 := IsDivisibleBy(m, ℓ);
    if ok then pp := [1] cat
      [r : r in [3..m0 by 2] |
        GCD(m0, r) eq 1 and m0 eq GCD(m0, (r+1) div 2)*GCD(m0, (r−1) div 2)];
    end if;
  else
    ok, n := IsDivisibleBy(2*m, ℓ);
```

```
    if ok then pp := [1] cat [r : r in [2..n div 2] |
      GCD(n,r) eq 1 and GCD(n div GCD(n,r+1), n div GCD(n,r-1)) eq 1];
    end if;
  end if;
  return pp;
end function;
```

Check that the arguments correspond to a quaternionic reflection group $G(\mathcal{D}_m, C_\ell, \psi_r)$ where $\mathcal{D}_m/C_\ell$ is either binary dihedral or dihedral.

```
qArgCheck := function(m,ℓ,r)
  ok, n := IsDivisibleBy(2*m,ℓ);
  if ok then
    ok := (r eq 1) or ((1 lt r) and (r le (n div 2)) and (GCD(n,r) eq 1)
      and (GCD(n div GCD(n,r+1), n div GCD(n,r-1)) eq 1));
  end if;
  return ok;
end function;
```

The group $G(\mathcal{D}_m, C_\ell, \psi_r)$.

```
qDC := function(m,ℓ,r)
  error if not qArgCheck(m,ℓ,r), "invalid argument(s)";
  n := 2*m div ℓ;
  K := qD(m);
  A := BaseRing(K);
  x := (K.1)[1,1]; y := (K.2)[1,1];
  gens := [Matrix(A,2,[0,1, 1,0]),Matrix(A,2,[x^n,0, 0,1]),Matrix(A,2,[x,0, 0,x^r]),
    Matrix(A,2,[y,0, 0,y^-1])];
  G := MatrixGroup< 2,A | gens >;
  assert #G eq 8*m*ℓ;
  return G;
end function;
```

A version of qDC that also returns a square root of 2.

```
a2DC := function(m,ℓ,r)
  error if not qArgCheck(m,ℓ,r), "invalid argument(s)";
  n := 2*m div ℓ;
  K, a2 := a2D(m);
  A := BaseRing(K);
  x := (K.1)[1,1]; y := (K.2)[1,1];
  gens := [Matrix(A,2,[0,1, 1,0]),Matrix(A,2,[x^n,0, 0,1]),Matrix(A,2,[x,0, 0,x^r]),
    Matrix(A,2,[y,0, 0,y^-1])];
  G := MatrixGroup< 2,A | gens >;
  assert #G eq 8*m*ℓ;
```

```
  return G, a2 ;
end function;
```

The group $G(\mathcal{D}_m, \mathcal{D}_\ell, 1)$ where m/ℓ is 1 or 2 and $\ell \geq 2$.

```
qDD := function(m, ℓ)
  error if ℓ le 1, "second argument must be at least 2";
  error if m notin [ℓ,2∗ℓ], "invalid arguments";
  D<x,y> := qD(m);
  A := BaseRing(D);
  refs := [ Matrix(A,2,[0,1, 1,0]), Matrix(A,2,[y[1,1],0,0,1]) ];
  if m eq ℓ then
    Append(~refs, Matrix(A,2, [x[1,1],0, 0,1]) );
  else
    Append(~refs, Matrix(A,2, [x[1,1],0, 0,x[1,1]^-1]) );
  end if;
  G := MatrixGroup< 2,A | refs >;
  assert #G eq 32∗m∗ℓ;
  return G;
end function;
```

4. The binary polyhedral groups $\mathcal{T}$, $\mathcal{O}$ and $\mathcal{I}$

For compatibility, the base ring of all three groups contains $a_2 = \sqrt{2}$ and $\tau = \frac{1}{2}(1+\sqrt{5})$.

The binary tetrahedral group $\mathcal{T} = \langle\, i, \omega \,\rangle$ of order 24, where $\omega = \frac{1}{2}(-1+i+j+k)$. Note that $\omega^{-1} i \omega = j$.

```
qT := function()
  _<t> := PolynomialRing(Rationals());
  K<a2,τ> := NumberField([t^2−2, t^2−t−1]);
  A<i,j,k> := QuaternionAlgebra<K | −1,−1>;
  return MatrixGroup< 1,A | [Matrix(A,1,[x]) : x in [A.1,(1/2)∗(−1+i+j+k)]] >;
end function;
```

The binary octahedral group $\mathcal{O} = \langle\, \frac{1}{2}(-1+i+j+k), \frac{1}{\sqrt{2}}(1+i) \,\rangle$ of order 48.

```
qO := function()
  _<t> := PolynomialRing(Rationals());
  K<a2,τ> := NumberField([t^2−2, t^2−t−1]);
  A<i,j,k> := QuaternionAlgebra<K | −1,−1>;
  ω := (1/2)∗(−1+i+j+k);
  γ := a2^-1∗(1+i);
  return MatrixGroup< 1,A | [Matrix(A,1,[x]) : x in [ω,γ]] >;
end function;
```

The binary icosahedral group $\mathcal{I} = \langle i, \frac{1}{2}(\tau^{-1} + i + \tau j) \rangle$ of order 120, where $\tau = \frac{1}{2}(1 + \sqrt{5})$.

```
qI := function()
  _<t> := PolynomialRing(Rationals());
  K<a2,τ> := NumberField([t^2−2, t^2−t−1]);
  A<i,j,k> := QuaternionAlgebra<K | −1, −1>;
  σ := (1/2)*(τ^−1 + i + τ*j);
  return MatrixGroup< 1, A | [Matrix(A,1,[x]) : x in [A.1, σ]] >;
end function;
```

5. The rank 2 quaternionic reflection groups $G(K, H, \varphi)$, $K \in \{\mathcal{T}, \mathcal{O}, \mathcal{I}\}$

The binary tetrahedral family

Let $\delta = \frac{1}{\sqrt{2}}(i - j)$ and let $\rho(\delta)$ denote conjugation by δ.

The group $G(\mathcal{T}, \mathcal{T}, \mathbf{1})$ of order $1\,152 = 2^7\,3^2$.

```
qTT := function()
  T := qT();
  A := BaseRing(T);
  zi := (T.1)[1,1];
  ω := (T.2)[1,1];
  zj := ω^−1*zi*ω;
  r1 := Matrix(A,2,[zi,0,0,1]);
  r2 := Matrix(A,2,[1,0,0,ω]);
  r3 := Matrix(A,2,[0,zj,zj^−1,0]);
  return MatrixGroup< 2,A | r1, r2, r3 >, {@ x : x in T @};
end function;
```

The group $G(\mathcal{T}, \mathcal{D}_2, \rho(\delta))$ of order $384 = 2^7\,3$.

```
qTD2d := function()
  T := qT();
  A<i,j,k> := BaseRing(T);
  K<a2,τ> := BaseRing(A);
  ω := 1/2*(−1+i+j+k);
  δ := 1/a2*(i−j);
  g1 := Matrix(A,2, [0,1, 1,0]);
  g2 := Matrix(A,2, [i,0, 0,1]);
  g3 := Matrix(A,2, [j,0, 0,1]);
  g4 := Matrix(A,2, [i,0, 0,−j]);
  g5 := Matrix(A,2, [ω,0, 0, δ*ω*δ^−1]);
  return MatrixGroup< 2,A | g1,g2,g3,g4,g5 >;
end function;
```

The group $G(\mathcal{T}, C_2, \rho(\delta)$ of order $96 = 2^5\,3$. It is isomorphic to $G(\mathcal{O}, \mathbf{1}, \beta)$ and $C_4 \boxdot_2 \mathcal{O}$.

```
qTC2d := function()
   T := qT();
   A<i,j,k> := BaseRing(T);
   K<a2,τ> := BaseRing(A);
   δ := 1/a2*(i−j);
   g1 := (T.1)[1,1];
   g2 := (T.2)[1,1];
   r1 := Matrix(A,2, [0,1, 1,0]);
   r2 := Matrix(A,2, [−1,0, 0,1]);
   r3 := Matrix(A,2, [g1,0, 0,δ*g1*δ^−1]);
   r4 := Matrix(A,2, [g2,0, 0,δ*g2*δ^−1]);
   return MatrixGroup< 2,A | r1,r2,r3,r4 >;
end function;
```

The group $G(\mathcal{T}, \mathbf{1}, \rho(\delta)) \simeq G_{12}$ of order $48 = 2^4\,3$ is the Shephard and Todd group G_{12}.

```
qTC1d := function()
   T := qT();
   A<i,j,k> := BaseRing(T);
   K<a2,τ> := BaseRing(A);
   δ := 1/a2*(i−j);
   g1 := (T.1)[1,1];
   g2 := (T.2)[1,1];
   r1 := Matrix(A,2, [0,1, 1,0]);
   r2 := Matrix(A,2, [g1,0, 0,δ*g1*δ^−1]);
   r3 := Matrix(A,2, [g2,0, 0,δ*g2*δ^−1]);
   return MatrixGroup< 2,A | r1,r2,r3 >;
end function;
```

The binary octahedral family

The group $G(\mathcal{O}, \mathcal{O}, \mathbf{1})$ of order 4 608.

```
qOO := function()
   O := qO();
   A := BaseRing(O);
   r1 := Matrix(A,2,[(O.1)[1,1],0,0,1]);
   r2 := Matrix(A,2,[(O.2)[1,1],0,0,1]);
   r3 := Matrix(A,2,[0,1, 1,0]);
   return MatrixGroup< 2,A | r1, r2, r3 >;
end function;
```

The group $G(\mathcal{O}, \mathcal{T}, \mathbf{1})$ of order 2 304.

```
qOT := function()
   O := qO();
   A := BaseRing(O);
   ω := (O.1)[1,1];
   γ := (O.2)[1,1];
   r1 := Matrix(A,2,[ω,0, 0,1]);
   r2 := Matrix(A,2,[0,1, 1,0]);
   r3 := Matrix(A,2,[0,ω^-1*γ*ω, ω^-1*γ^-1*ω,0]);
   return MatrixGroup< 2,A | r1, r2, r3 >;
end function;
```

The group $G(\mathcal{O}, \mathcal{D}_2, \mathbf{1})$ of order 768.

```
qOD2 := function()
   O := qO();
   A<i,j,k> := BaseRing(O);
   γ := (O.2)[1,1];
   K<a2,τ> := BaseRing(A);
   r1 := Matrix(A,2,[γ^2,0, 0,1]);
   r2 := Matrix(A,2,[0,1, 1,0]);
   r3 := Matrix(A,2,[0,x,x^-1,0]) where x is 1/a2*(j+k);
   r4 := Matrix(A,2,[0,x,x^-1,0]) where x is 1/a2*(1+j);
   D2 := sub< O | O.2^2, O.1^-1*O.2^2*O.1 >;
   return MatrixGroup< 2,A | r1,r2,r3,r4 >;
end function;
```

The group $G(\mathcal{O}, C_2, \mathbf{1})$ of order 192. It is isomorphic to $C_4 \boxdot \mathcal{O}$.

```
qOC2 := function()
   O := qO();
   A<i,j,k> := BaseRing(O);
   K<a2,τ> := BaseRing(A);
   g1 := (O.1)[1,1];
   g2 := (O.2)[1,1];
   r1 := Matrix(A,2, [0,1, 1,0]);
   r2 := Matrix(A,2, [-1,0, 0,1]);
   r3 := Matrix(A,2, [g1,0, 0,g1]);
   r4 := Matrix(A,2, [g2,0, 0,g2]);
   return MatrixGroup< 2,A | r1,r2,r3,r4 >;
end function;
```

The group $G(\mathcal{O}, \mathbf{1}, \beta)$ of order 96. It is isomorphic to $G(\mathcal{T}, C_2, \rho(\delta))$ and $C_4 \boxdot_2 \mathcal{O}$.

```
qOC1b := function()
  O := qO();
  A<i,j,k> := BaseRing(O);
  K<a2,τ> := BaseRing(A);
  g1 := (O.1)[1,1];
  g2 := (O.2)[1,1];
  β := hom< O → O | O.1, −O.2 >;    // central automorphism
  r1 := Matrix(A,2, [0,1, 1,0]);
  r2 := Matrix(A,2, [g1,0, 0,g1]);
  r3 := Matrix(A,2, [g2,0, 0,−g2]);
  return MatrixGroup< 2,A | r1,r2,r3 >, β;
end function;
```

The group $G(\mathcal{O}, \mathbf{1}, \rho(\delta))$ of order 96 is the Shephard and Todd group G_{13}.

```
qOC1d := function()
  O := qO();
  A<i,j,k> := BaseRing(O);
  K<a2,τ> := BaseRing(A);
  δ := 1/a2∗(i−j);
  g1 := (O.1)[1,1];
  g2 := (O.2)[1,1];
  r1 := Matrix(A,2, [0,1, 1,0]);
  r2 := Matrix(A,2, [g1,0, 0,δ∗g1∗δ^−1]);
  r3 := Matrix(A,2, [g2,0, 0,δ∗g2∗δ^−1]);
  return MatrixGroup< 2,A | r1,r2,r3 >;
end function;
```

The binary icosahedral family

The group $G(\mathcal{I}, \mathcal{I}, \mathbf{1})$ of order 28 800.

```
qII := function()
  I := qI();
  A<i,j,k> := BaseRing(I);
  r1 := Matrix(A,2,[0,1,1,0]);
  r2 := Matrix(A,2,[1,0,0,i]);
  r3 := Matrix(A,2,[1,0,0,(I.2)[1,1]]);
  return MatrixGroup< 2,A | r1, r2, r3 >;
end function;
```

The group $G(\mathcal{I}, C_2, \mathbf{1})$ of order 480. It is isomorphic to $C_4 \boxdot \mathcal{I}$.

```
qIC2 := function()
  I := qI();
  A<i,j,k> := BaseRing(I);
  K<a2,τ> := BaseRing(A);
  g1 := (I.1)[1,1];
  g2 := (I.2)[1,1];
  r1 := Matrix(A,2, [0,1, 1,0]);
  r2 := Matrix(A,2, [−1,0, 0,1]);
  r3 := Matrix(A,2, [g1,0, 0,g1]);
  r4 := Matrix(A,2, [g2,0, 0,g2]);
  return MatrixGroup< 2,A | r1,r2,r3,r4 >;
end function;
```

The group $G(\mathcal{I}, C_2, \Theta)$ of order 480.

```
qIC2t := function()
  I := qI();
  A<i,j,k> := BaseRing(I);
  K<a2,τ> := BaseRing(A);
  M := Matrix(A,1,[−1/2*τ − 1/2*i + 1/2*(τ − 1)*k]);
  θ := hom< I → I | −I.1, M >;
  g1 := (I.1)[1,1];
  g2 := (I.2)[1,1];
  r1 := Matrix(A,2, [0,1, 1,0]);
  r2 := Matrix(A,2, [−1,0, 0,1]);
  r3 := Matrix(A,2, [g1,0, 0,−g1]);
  r4 := Matrix(A,2, [g2,0, 0,M[1,1]]);
  return MatrixGroup< 2,A | r1,r2,r3,r4 >, θ;
end function;
```

The group $G(\mathcal{I}, \mathbf{1}, \Theta)$ of order 240.

```
qIC1t := function()
  I := qI();
  A<i,j,k> := BaseRing(I);
  K<a2,τ> := BaseRing(A);
  δ := 1/a2*(i−j);
  M := Matrix(A,1,[−1/2*τ − 1/2*i + 1/2*(τ − 1)*k]);
  θ := hom< I → I | −I.1, M >;
  g1 := (I.1)[1,1];
  g2 := (I.2)[1,1];
  r1 := Matrix(A,2, [0,1, 1,0]);
  r2 := Matrix(A,2, [g1,0, 0,−g1]);
```

```
    r3 := Matrix(A, 2, [g2, 0, 0, M[1, 1]]);
    return MatrixGroup< 2, A | r1, r2, r3 >, θ;
  end function;
```

The group $G(\mathcal{I}, \mathbf{1}, \rho(j))$ of order 240 is the Shephard and Todd group G_{22}.

```
  qIC1j := function()
    I := qI();
    A<i, j, k> := BaseRing(I);
    K<a2, τ> := BaseRing(A);
    δ := 1/a2∗(i−j);
    g1 := (I.1)[1, 1];
    g2 := (I.2)[1, 1];
    r1 := Matrix(A, 2, [0, 1, 1, 0]);
    r2 := Matrix(A, 2, [g1, 0, 0, j∗g1∗j^−1]);
    r3 := Matrix(A, 2, [g2, 0, 0, j∗g2∗j^−1]);
    return MatrixGroup< 2, A | r1, r2, r3 >;
  end function;
```

6. Primitive rank 2 quaternionic reflection groups with imprimitive complexification

Let K be a non-cyclic binary polyhedral subgroup and let H be a normal subgroup of K. If there is a surjective homomorphism $\psi : C_d \to K/H$ and $f = |K/H|$, let $C_d \times_f K$ denote the pullback

$$\begin{array}{ccc} C_d \times_f K & \xrightarrow{\pi_K} & K \\ \Big\downarrow{\scriptstyle \pi_d} & & \Big\downarrow{\scriptstyle \mu} \\ C_d & \xrightarrow{\psi} & K/H \end{array}$$

For $\alpha \in C_d$, $\xi \in K$ the map $\theta(\alpha, \xi)$ that sends $h \in \mathbb{H}$ to $\alpha h \overline{\xi}$ is a $\mathbb{C}$-linear transformation of $\mathbb{H}$ with kernel $\{\pm(1, 1)\}$. The image $C_d \circ_f K = \theta(\mu_d \times_f K)$ is an irreducible subgroup of $\mathrm{U}_2(\mathbb{C})$ of order $\dfrac{d}{2f}|K|$. (Omit f when $f = 1$.)

Let $s = \begin{pmatrix} 0, j \\ -j, 0 \end{pmatrix}$. If K is $\mathcal{T}$, $\mathcal{O}$ or $\mathcal{I}$ and d is even, all groups $C_d \circ_f K$ except $C_d \circ_3 \mathcal{T}$ are normalised by s. Define $C_d \boxdot_f K = (C_d \circ_f K)\langle s \rangle$.

This is $C_d \boxdot \mathcal{T}$ for $6 \mid d$. Order $24d$.

```
  qmuT := function(d)
    error if d mod 6 ne 0, "d must be divisible by 6";
    μ := qC(d);
    A<i, j, k> := BaseRing(μ);
    g0 := Matrix(A, 2, [z, 0, 0, z]) where z is (μ.1)[1, 1];
    g1 := Matrix(A, 2, [i, 0, 0, −i]);
    g2 := 1/2∗Matrix(A, 2, [−1+i, 1+i, −1+i, −1−i]);
```

```
    s := Matrix(A, 2, [0, −j, j, 0] );
    return MatrixGroup< 2, A | g0, g1, g2, s >;
end function;
```

This is $C_d \boxdot \mathcal{O}$ for $4 \mid d$. Order $48d$. It is primitive unless $d = 4$.

```
qmuO := function(d)
    error if d mod 4 ne 0, "d must be divisible by 4";
    μ, a2 := a2C(d);
    A<i, j, k> := BaseRing(μ);
    g0 := Matrix(A, 2, [z, 0, 0, z]) where z is (μ.1)[1, 1];
    g1 := 1/2 ∗ Matrix(A, 2, [−1+i, 1+i, −1+i, −1−i]);
    g2 := Matrix(A, 2, [κ, 0, 0, κ^−1]) where κ is 1/a2∗(1+i);
    s := Matrix(A, 2, [0, −j, j, 0] );
    return MatrixGroup< 2, A | g0, g1, g2, s >;
end function;
```

This is $C_d \boxdot_2 \mathcal{O}$ for $4 \mid d$ and $16 \nmid d$. Order $24d$. It is primitive unless $d = 4$.

```
qmu2O := function(d)
    error if d mod 4 ne 0, "d must be a multiple of 4";
    error if d mod 16 eq 0, "d must not be a multiple of 16";
    μ, a2 := a2C(d);
    A<i, j, k> := BaseRing(μ);
    g0 := Matrix(A, 2, [z, 0, 0, z]) where z is (μ.1)[1, 1];
    g1 := 1/2 ∗ Matrix(A, 2, [−1+i, 1+i, −1+i, −1−i]);
    g2 := Matrix(A, 2, [κ, 0, 0, κ^−1]) where κ is 1/a2∗(1+i);
    s := Matrix(A, 2, [0, −j, j, 0] );
    return MatrixGroup< 2, A | g0^2, g1, g2^2, g0∗g2, s >;
end function;
```

This is $C_d \boxdot \mathcal{I}$ for d divisible by 4, 6 or 10. Order $120d$. It is primitive unless $d = 4$.

```
qmuI := function(d)
    error if (d mod 4) ne 0 and (d mod 6) ne 0 and (d mod 10) ne 0,
      "d must be divisible by 4, 6, or 10";
    μ, a5 := a5C(d);
    A<i, j, k> := BaseRing(μ);
    τ := 1/2∗(1+a5);
    g0 := Matrix(A, 2, [z, 0, 0, z]) where z is (μ.1)[1, 1];
    g1 := 1/2∗Matrix(A, 2, [τ^−1 + τ∗i, 1, −1, τ^−1 − τ∗i]);
    g2 := Matrix(A, 2, [i, 0, 0, −i]);
    s := Matrix(A, 2, [0, −j, j, 0] );
    return MatrixGroup< 2, A | g0, g1, g2, s >;
end function;
```

7. Reflections

The standard quaternionic hermitian inner product.

```
qDotProduct := function(u, v)
  error if v notin Generic(Parent(u)), "vectors must be in the same space";
  vbar := Vector([Conjugate(x) : x in Eltseq(v)]);
  return InnerProduct(u, vbar);
end function;
```

The reflection with root a and eigenvalue ξ.

```
qReflection := function(a, ξ)
  V := Generic(Parent(a));
  A := BaseRing(V);
  λ := (qDotProduct(a, a))^-1 * (One(A) - ξ);
  return Matrix([ v - qDotProduct(v, a)*λ*a : v in Basis(V)]);
end function;
```

The indexed set of all reflections in G.

```
reflections := func< G |
  &join[Class(G, c[3]) : c in Classes(G) | Rank(Matrix(c[3])-G.0) eq 1 ] >;
```

8. Irreducible imprimitive quaternionic reflection groups of rank ≥ 3

If G is a finite irreducible imprimitive quaternionic reflection group of rank $n \geq 3$, there is a unique binary polyhedral group K and a normal subgroup $H \trianglelefteq K$ with $[K, K] \subseteq H$ such that G is conjugate to

$$G_n(K, H) = \{ \operatorname{diag}(\theta_1, \theta_2, \ldots, \theta_n) \mid \theta_r \in K \text{ and } \theta_1\theta_2 \cdots \theta_n \in H \} \rtimes \operatorname{Sym}(n).$$

The order of $G_n(K, H)$ is $|K|^{n-1}\,|H|n!$.

The group $G_n(K, H)$.

```
qImprimitive := function(K, H, n)
  error if not DerivedGroup(K) subset H, "K/H must be abelian";
  A := BaseRing(K);
  gens := [ PermutationMatrix(A, g) : g in Generators(Sym(n)) ];
  M := RSpace(A, n);
  gens cat:= [ qReflection(M.1, ξ[1, 1]) : ξ in Generators(H) ];
  gens cat:= [ qReflection(M.1 - ξ[1, 1]*M.2, -Id(A)) : ξ in Generators(K) ];
  return MatrixGroup< n, A | gens >;
end function;
```

9. Quaternionic reflection groups with primitive complexification

Rank 2

Double cover of Alt(5).

```
qqO_1 := function()
  _<t> := PolynomialRing(Rationals());
  K<a,b> := NumberField([t^2-2, t^2-3]);
  A<ii,jj,kk> := QuaternionAlgebra< K | -1, -1 >;
  γ := 1/2*(-1 + b*ii);
  δ := 1/2*(-1 + b*jj);
  V := RSpace(A,2);
  v_1 := elt< V | 0, γ - 1 >;
  v_2 := elt< V | a*jj, γ >;
  v_3 := elt< V | -x*a^-1*δ*jj, x*γ^2 > where x is (1-kk);
  rts := [v_1, v_2, v_3];
  refs := [ qReflection(v,γ) : v in rts ];
  return MatrixGroup< 2,A | refs >, rts;
end function;
```

Double cover of Alt(6).

```
qqO_2 := function()
  _<t> := PolynomialRing(Rationals());
  K<a,b> := NumberField([t^2-2, t^2-3]);
  A<ii,jj,kk> := QuaternionAlgebra< K | -1, -1 >;
  γ := 1/2*(-1 + b*ii);
  δ := 1/2*(-1 + b*jj);
  V := RSpace(A,2);
  v_1 := elt< V | 0, γ - 1 >;
  v_2 := elt< V | a*jj, γ >;
  v_3 := elt< V | -x*a^-1*δ*jj, x*γ^2 > where x is (1-kk);
  v_4 := elt< V | a*b*(1-kk), 0 >;
  rts := [v_1, v_2, v_3, v_4];
  refs := [ qReflection(v,γ) : v in rts ];
  return MatrixGroup< 2,A | refs >;
end function;
```

Double cover of Sym(6).

```
qqO_3 := function()
  _<t> := PolynomialRing(Rationals());
  K<a,b,c> := NumberField([t^2-2, t^2-3, t^2-5]);
  A<ii,jj,kk> := QuaternionAlgebra< K | -1, -1 >;
```

```
γ := 1/2*(-1 + b*ii);
τ := 1/2*(2 + b + ii + b*jj - kk);
V := RSpace(A,2);
v1 := elt< V | a*jj, 1 + b >;
v2 := elt< V | γ*a*jj, 1 + b >;
v3 := elt< V | a*jj, γ*(1 + b) >;
v4 := elt< V | τ*γ*(1+kk)/a, τ*γ >;
v5 := elt< V | x*γ^2*(1-kk)/a, -x*γ > where x is ii*τ;
rts := [v1, v2, v3, v4, v5];
refs := [ qReflection(v, -1) : v in rts ];
return MatrixGroup< 2, A | refs >;
end function;
```

Extensions of the extraspecial group $\mathbf{2}_{-}^{1+4}$

The extraspecial group $\mathbf{2}_{-}^{1+4}$ is the central product $Q_8 \circ D_8$ of order 32, where $Q_8 = \mathcal{D}_2$ is the quaternion group of order 8 and D_8 is the dihedral group of order 8.

The group $E\mathcal{D}_5$ where $E = \mathbf{2}_{-}^{1+4}$ is a normal subgroup, $\mathcal{D}_5$ is the binary dihedral group of order 20 and $E \cap \mathcal{D}_5 = Z(E)$. There are 20 reflections of order 4, 10 of order 2 and 10 root lines. The order of $E\mathcal{D}_5$ is 320.

```
qqP1 := function()
  A<ii,jj,kk> := QuaternionAlgebra<Rationals() | -1,-1>;
  V := RSpace(A,2);
  v1 := elt< V | 1,0>;
  v2 := elt< V | x,x > where x is 1/2*(1-ii);
  rts := [v1, v2];
  refs := [ qReflection(v, jj) : v in rts ];
  return MatrixGroup< 2, A | refs >, rts;
end function;
```

The group EA where $E = \mathbf{2}_{-}^{1+4}$, $A = \mathrm{SL}(2,5)$ and $E \cap A = Z(E)$. There are 60 reflections of order 4, 10 of order 2. The order of ES is 1 920.

```
qqP2 := function()
  A<ii,jj,kk> := QuaternionAlgebra<Rationals() | -1,-1>;
  V := RSpace(A,2);
  v1 := elt< V | 1,0>;
  v2 := elt< V | 0,1>;
  v3 := elt< V | x,x*jj > where x is 1/2*(1-ii);
  r1 := qReflection(v1, kk);
  r2 := qReflection(v2, jj);
  r3 := qReflection(v3, kk);
  return MatrixGroup< 2, A | [r1, r2, r3 ] >, [v1,v2,v3];
end function;
```

The group ES where $E = 2_-^{1+4}$ is a normal subgroup, S is a double cover of $\mathrm{Sym}(5)$ and $E \cap S = Z(E)$. There is a single orbit of 60 reflections of order 4. There are 50 reflections of order 2: an orbit of length 40 and an orbit of length 10. The order of ES is 3 840.

```
qqP3 := function()
  _<t> := PolynomialRing(Rationals());
  K<a> := NumberField(t^2-2);
  A<ii,jj,kk> := QuaternionAlgebra< K | -1, -1 >;
  V := RSpace(A,2);
  v1 := elt< V | 1+a, ii>;
  v2 := elt< V | 1+a, jj>;
  v3 := elt< V | 1+a, kk>;
  v4 := elt< V | x/a*(1+jj), x*jj> where x is 1 + a^-1*(1+jj);
  v5 := elt< V | x/a*(1-kk), x*kk> where x is 1 + a^-1*(1+kk);
  rts := [v1, v2, v3, v4, v5];
  refs := [ qReflection(v, -1) : v in rts ];
  return MatrixGroup< 2, A | refs >;
end function;
```

Rank 3

The group $\mathbb{Z}_2 \times \mathrm{SU}(3,3)$ of order 12 096.

```
qqQ := function()
  _<t> := PolynomialRing(Rationals());
  K<c> := NumberField(t^2-5);
  A<ii,jj,kk> := QuaternionAlgebra< K | -1, -1 >;
  V := RSpace(A,3);
  α := 2^-1*(1-ii-jj-c*kk);
  v1 := elt< V | 2,0,0 >;
  v2 := elt< V | 0,2,0 >;
  v3 := elt< V | ii, ii, α>;
  v4 := elt< V | α*ii, 1, ii>;
  rts := [v1, v2, v3, v4];
  refs := [ qReflection(v, -1) : v in rts ];
  return MatrixGroup< 3, A | refs >;
end function;
```

The double cover of the Janko group J_2. Its order is 1 209 600. It contains 315 reflections. They form a single conjugacy class of elements of order 2.

```
qqR := function()
  _<t> := PolynomialRing(Rationals());
  K<τ> := NumberField(t^2-t-1);
```

```
    A<ii,jj,kk> := QuaternionAlgebra< K | −1, −1 >;
    V := RSpace(A,3);
    c := 2∗τ − 1;
    σ := 1 − τ;
    α := 2^−1∗(1−ii−jj−c∗kk);
    ω := 1/2∗(−1+ii+jj+kk);
    v1 := elt< V | 2,0,0 >;
    v2 := elt< V | 0,2,0 >;
    v3 := elt< V | ii,ii,α>;
    v4 := elt< V | τ∗ω^2, σ∗ω∗jj,1>;
    rts := [v1,v2,v3,v4];
    refs := [ qReflection(v,−1) : v in rts ];
    return MatrixGroup< 3,A | refs >;
end function;
```

Rank 4

The split extension of $Q_8 \circ D_8 \circ D_8$ by the Shephard and Todd group $G(3,3,3)$.

```
qqS1 := function()
    A<ii,jj,kk> := QuaternionAlgebra<Rationals() | −1,−1>;
    V := RSpace(A,4);
    v1 := elt< V | x,x,0,0> where x is 1 − ii;
    v2 := elt< V | x,0,x,0> where x is 1 − ii;
    v3 := elt< V | x,0,0,x> where x is 1 − ii;
    v4 := elt< V | 1,ii,jj,kk >;
    rts := [v1,v2,v3,v4];
    refs := [ qReflection(v,−1) : v in rts ];
    return MatrixGroup< 4,A | refs >;
end function;
```

The split extension of $Q_8 \circ D_8 \circ D_8$ by the Shephard and Todd group $G(3,3,4)$.

```
qqS2 := function()
    A<ii,jj,kk> := QuaternionAlgebra<Rationals() | −1,−1>;
    V := RSpace(A,4);
    v1 := elt< V | x,x,0,0> where x is 1 − ii;
    v2 := elt< V | x,0,0,x> where x is 1 − ii;
    v3 := elt< V | 1,ii,jj,kk >;
    v4 := elt< V | 0,0,0,2>;
    rts := [v1,v2,v3,v4];
    refs := [ qReflection(v,−1) : v in rts ];
    return MatrixGroup< 4,A | refs >;
end function;
```

The non-split extension of $Q_8 \circ D_8 \circ D_8$ by $\Omega^-(6,2)$.

```
qqS3 := function()
  A<ii,jj,kk> := QuaternionAlgebra<Rationals() | -1,-1>;
  V := RSpace(A,4);
  v1 := elt< V | x, x*ii, 0, 0> where x is 1 - ii;
  v2 := elt< V | x, x*jj, 0, 0> where x is 1 - ii;
  v3 := elt< V | x, 0, x*kk, 0> where x is 1 - ii;
  v4 := elt< V | x*ii, 0, x, 0> where x is 1 - ii;
  v5 := elt< V | 1, jj, kk, ii >;
  rts := [v1, v2, v3, v4, v5];
  refs := [ qReflection(v, -1) : v in rts ];
  return MatrixGroup< 4, A | refs >;
end function;
```

Example

```
G := qqS3();
E := pCore(G);
Q := quo< G | E >;
ok := IsIsomorphic(Q, OmegaMinus(6,2)); ok;
```

The split extension of $\mathrm{SL}(2,5) \circ \mathrm{SL}(2,5) \circ \mathrm{SL}(2,5)$ by $\mathrm{Sym}(3)$.

```
qqT := function()
  _<t> := PolynomialRing(Rationals());
  K<τ> := NumberField(t^2-t-1);
  A<ii,jj,kk> := QuaternionAlgebra< K | -1, -1 >;
  V := RSpace(A,4);
  σ := 1 - τ;
  ω := 1/2*(-1+ii+jj+kk);
  p := 1/2*(τ + σ*ii - jj);
  v1 := elt< V | 1/2*τ, 1/2*σ, -1/2, 0 >;
  v2 := elt< V | 1/2*σ, 1/2*τ, 0, -1/2 >;
  v3 := elt< V| 1, 0, 0, 0 >;
  v4 := elt< V| x, x*p^-1*ii*p, x*p^-1*jj*p, x*p^-1*kk*p> where x is 1/2;
  rts := [v1, v2, v3, v4];
  refs := [ qReflection(v, -1) : v in rts ];
  return MatrixGroup< 4, A | refs >;
end function;
```

Rank 5

The group $\mathbb{Z}_2 \times \mathrm{SU}(5,2)$.

```
qqU := function()
  _<t> := PolynomialRing(Rationals());
  K<τ> := NumberField(t^2-t-1);
  A<ii,jj,kk> := QuaternionAlgebra< K | -1, -1 >;
  V := RSpace(A,5);
  ω := 1/2*(-1+ii+jj+kk);
  v1 := elt< V| 2,0,0,0,0>;
  v2 := elt< V| 0,0,0,2,0>;
  v3 := elt<V | 0,1,ω,ω,1>;
  v4 := elt<V | 1,0,1,ω,ω>;
  v5 := elt< V | 0,1,ii*ω*ii,jj*ω*jj,ii >;
  rts := [v1,v2,v3,v4,v5];
  refs := [ qReflection(v,-1) : v in rts ];
  return MatrixGroup< 5,A | refs >;
end function;
```

10. Test code

```
load "QuatRefCode.m";

"Checking qDC(m,ell,r) for m in [1..12]";
time for m := 1 to 12 do
  for ℓ in Divisors(2*m) do
    for r in qValidParametersDC(m,ℓ) do
      assert #qDC(m,ℓ,r) eq 8*m*ℓ;
end for; end for; end for;

"Checking qDD(m,ell) for m in [2..12]";
time for m := 2 to 12 do
  assert #qDD(m,m) eq 32*m^2;
  assert #qDD(2*m,m) eq 64*m^2;
end for;

gpsT := [qTT,qTD2d,qTC2d,qTC1d];
ordT := [1152,384,96,48];
gpsO := [qOO,qOT,qOD2,qOC2,qOC1b,qOC1d];
ordO := [4608,2304,768,192,96,96];
gpsI := [qII,qIC2,qIC2t,qIC1t,qIC1j];
ordI := [28800,480,480,240,240];

"Checking binary tetrahedral types";
time for n → gp in gpsT do assert #gp() eq ordT[n]; end for;

"Checking binary octahedral types";
time for n → gp in gpsO do assert #gp() eq ordO[n]; end for;
```

```
"Checking binary icosahedral types";
time for n → gp in gpsI do assert #gp() eq ordI[n]; end for;

"Checking qmuT";
time for d in [6,12,18,24] do assert #qmuT(d) eq 24*d; end for;

"Checking qmuO";
time for d in [4..24 by 4] do assert #qmuO(d) eq 48*d; end for;

"Checking qmu2O";
time for d in [4,8,12,20,24] do assert #qmu2O(d) eq 24*d; end for;

"Checking qmuI";
time for d in [4,6,8,10,12] do assert #qmuI(d) eq 120*d; end for;

"Checking G_n(T,T) for n in [2..8]";
T := qT();
time for n := 2 to 8 do #qImprimitive(T, T, n) eq #T^n*Factorial(n); end for;

"Checking G_n(O,T) for n in [2..8]";
O := qT(); T := sub< O | O.1, O.2^2 >;
time for n := 2 to 8 do #qImprimitive(O, T, n) eq #O^(n-1)*#T*Factorial(n); end for;

"Group  order #reflections";
gpsPrim := [qqO1, qqO2, qqO3, qqP1, qqP2, qqP3, qqQ, qqR, qqS1, qqS2, qqS3, qqT, qqU];
gpName := ["W(O1)", "W(O2)", "W(O3)", "W(P1)", "W(P2)", "W(P3)",
    "W(Q) ", "W(R) ", "W(S1) ", "W(S2)", "W(S3)", "W(T) ", "W(U) "];
time for n → gp in gpsPrim do
   G := gp();
   rr := reflections(G);
   n, gpName[n], #G, #rr;
end for;
```